Problem:
Score : 400 points

Problem Statement

Takahashi has a maze, which is a grid of H \times W squares with H horizontal rows and W vertical columns.

The square at the i-th row from the top and the j-th column is a "wall" square if S_{ij} is #, and a "road" square if S_{ij} is ..

From a road square, you can move to a horizontally or vertically adjacent road square.

You cannot move out of the maze, move to a wall square, or move diagonally.

Takahashi will choose a starting square and a goal square, which can be any road squares, and give the maze to Aoki.

Aoki will then travel from the starting square to the goal square, in the minimum number of moves required.

In this situation, find the maximum possible number of moves Aoki has to make.

Constraints

1 \leq H,W \leq 20

S_{ij} is . or #.

S contains at least two occurrences of ..

Any road square can be reached from any road square in zero or more moves.

Input

Input is given from Standard Input in the following format:

H W
S_{11}...S_{1W}
:
S_{H1}...S_{HW}

Output

Print the maximum possible number of moves Aoki has to make.

Sample Input 1

3 3
...
...
...

Sample Output 1

4

If Takahashi chooses the top-left square as the starting square and the bottom-right square as the goal square, Aoki has to make four moves.

Sample Input 2

3 5
...#.
.#.#.
.#...

Sample Output 2

10

If Takahashi chooses the bottom-left square as the starting square and the top-right square as the goal square, Aoki has to make ten moves.

Solution in Haskell (GHC 8.8.3) (judge verdict: Accepted)

import qualified Data.Text.Lazy as T
import qualified Data.Text.Lazy.IO as T
import qualified Data.Text.Lazy.Read as T
import qualified Data.Vector.Unboxed as VU
import Data.IntSet (IntSet)
import qualified Data.IntSet as IS

main :: IO ()
main = do
  [h, w] <- map unsafeTextToInt . T.words <$> T.getLine :: IO [Int]
  let
    encode :: (Int, Int) -> Int
    encode (i, j) = i * w + j
    decode :: Int -> (Int, Int)
    decode = (`divMod` w)
    inside :: (Int, Int) -> Bool
    inside (i, j) = 0 <= i && i < h && 0 <= j && j < w
    neighbors :: VU.Vector Bool -> Int -> [Int]
    neighbors v k | v VU.! k  = filter (v VU.!) . map encode . filter inside $ [(i, j - 1), (i, j + 1), (i - 1, j), (i + 1, j)]
                  | otherwise = []
      where
        (i, j) = decode k
    search :: VU.Vector Bool -> Int -> Int
    search v k0 = snd . last $ queue
      where
        queue = (k0, 0) : walk 1 (IS.singleton k0) queue
        walk :: Int -> IntSet -> [(Int, Int)] -> [(Int, Int)]
        walk 0 _ _ = []
        walk l u ((k, d) : q) = map (flip (,) (d + 1)) ns ++ walk (l - 1 + length ns) u' q
          where
            ns = filter (`IS.notMember` u) $ neighbors v k
            u' = foldl (flip IS.insert) u ns
  bs <- VU.fromList . map ('.' ==) . concat . lines <$> getContents :: IO (VU.Vector Bool)
  print . VU.maximum . VU.map (search bs) $ VU.enumFromN 0 (h * w)

unsafeTextToInt :: T.Text -> Int
unsafeTextToInt s = case T.signed T.decimal s of
  Right (n, _) -> n
{-# INLINE unsafeTextToInt #-}